MODULE *HKFM*

EXTENDS *Integers*, *Sequences*
CONSTANTS *Client*, *Song*
VARIABLES *inbox*, *state*

Type definitions (kinda sorta) and other useful stuff...

There are various places where we want to refer to all variables at once, so it's useful to define a *vars* tuple.

$$vars \triangleq \langle inbox, state \rangle$$

The constant *Client* is the set of all clients, represented however we want. It's defined externally, in the model.

We define *Node* to be the set of all nodes in the system, including the server. We don't care how the server is represented, only that it doesn't clash with any of the clients. We use the TLA+ CHOOSE operator to express this.

$$\begin{aligned} Server &\triangleq \text{CHOOSE } x : x \notin Client \\ Node &\triangleq Client \cup \{Server\} \end{aligned}$$

These terms all relate to the *playhead*. A *playlist* has two fields:

i: the current track in the *playlist*
t: the number of seconds into that track

When $i = -1$ it means we're not playing anything.

Every node has a *State* consisting of two fields:

playlist: a sequence of songs from the constant set *Song*
playhead: as described above

$$\begin{aligned} Idx &\triangleq Nat \cup \{-1\} \\ Playlist &\triangleq Seq(Song) \\ Playhead &\triangleq [i : Idx,\ t : Nat] \\ Stopped &\triangleq [i \mapsto -1,\ t \mapsto 0] \\ State &\triangleq [playlist : Playlist,\ playhead : Playhead] \\ InitState &\triangleq [playlist \mapsto \langle\rangle,\ playhead \mapsto Stopped] \end{aligned}$$

Clients send "add", "seek", and "skip" messages to the server and the server sends "sync" messages to all clients whenever its state changes. The term *Message* is the set of all possible messages that can occur.

$$\begin{aligned} Message \triangleq\ & [action : \{\text{"sync"}\},\ data : State] \cup \\ & [action : \{\text{"add"}\},\ data : Song,\ sender : Client] \cup \\ & [action : \{\text{"seek"}, \text{"skip"}\},\ data : Playhead,\ sender : Client] \end{aligned}$$

The *TypeOK* formula states that *inbox* must be a function from nodes to sequences of messages and state must be a function from nodes to states. We can ask *TLC* to check that *TypeOK* is an invariant of every behaviour, meaning it will find circumstances where *inbox* and state end up looking wonky. It's also useful to have as a high level type definition for these variables.

$$\begin{aligned} TypeOK \triangleq\ & \land inbox \in [Node \to Seq(Message)] \\ & \land state\ \in [Node \to State] \end{aligned}$$

Message Constructors

These operators are just for convenience when creating messages in actions below.

$$SyncMsg \triangleq$$
$$[action \mapsto \text{“sync”},\ data \mapsto state'[Server]]$$

$$AddMsg(client,\ song) \triangleq$$
$$[action \mapsto \text{“add”},\ data \mapsto song,\ sender \mapsto client]$$

$$SeekMsg(client,\ playhead) \triangleq$$
$$[action \mapsto \text{“seek”},\ data \mapsto playhead,\ sender \mapsto client]$$

$$SkipMsg(client,\ playhead) \triangleq$$
$$[action \mapsto \text{“skip”},\ data \mapsto playhead,\ sender \mapsto client]$$

Client Actions

$$\begin{aligned} & SendAdd(self,\ song) \triangleq \\ & \quad \text{LET} \\ & \qquad msg \triangleq AddMsg(self,\ song) \\ & \quad \text{IN} \\ & \qquad \land inbox' = [inbox\ \text{EXCEPT}\ ![Server] = Append(inbox[Server],\ msg)] \\ & \qquad \land \text{UNCHANGED}\ state \end{aligned}$$

$$\begin{aligned} & RecvSync(self) \triangleq \\ & \quad \land inbox[self] \neq \langle\rangle \\ & \quad \land \text{LET} \\ & \qquad msg \triangleq Head(inbox[self]) \\ & \qquad tail \triangleq Tail(inbox[self]) \\ & \quad \text{IN} \\ & \qquad \land msg.action = \text{“sync”} \\ & \qquad \land inbox' = [inbox\ \text{EXCEPT}\ ![self] = tail] \\ & \qquad \land state'\ = [state\ \text{EXCEPT}\ ![self]\ = msg.data] \end{aligned}$$

$$\begin{aligned} & SendSeek(self) \triangleq \\ & \quad \text{LET} \\ & \qquad playhead \triangleq state[self].playhead \\ & \qquad msg \triangleq SeekMsg(self,\ [playhead\ \text{EXCEPT}\ !.t = playhead.t + 1]) \end{aligned}$$

```
  IN
    ∧ playhead ≠ Stopped
    ∧ inbox' = [inbox EXCEPT ![Server] = Append(inbox[Server], msg)]
    ∧ UNCHANGED state

SendSkip(self) ≜
  LET
    playhead ≜ state[self].playhead
    msg ≜ SkipMsg(self, playhead)
  IN
    ∧ playhead ≠ Stopped
    ∧ inbox' = [inbox EXCEPT ![Server] = Append(inbox[Server], msg)]
    ∧ UNCHANGED state
```

Server Actions

```
BroadcastSync ≜
  ∧ inbox' = [n ∈ Node ↦ IF n = Server
                           THEN Tail(inbox[n])
                           ELSE Append(inbox[n], SyncMsg)]

RecvAdd ≜
  ∧ inbox[Server] ≠ ⟨⟩
  ∧ LET
      server ≜ state[Server]
      msg ≜ Head(inbox[Server])
    IN
      ∧ msg.action = "add"
      ∧ LET
          newPlaylist  ≜ Append(server.playlist, msg.data)
          newPlayhead ≜ IF server.playhead = Stopped
                        THEN [i ↦ Len(server.playlist), t ↦ 0]
                        ELSE server.playhead
        IN
          ∧ state' = [state EXCEPT ![Server] = [playlist  ↦ newPlaylist,
                                                playhead ↦ newPlayhead]]
          ∧ BroadcastSync

RecvSeek ≜
  ∧ inbox[Server] ≠ ⟨⟩
  ∧ LET
      server ≜ state[Server]
      msg ≜ Head(inbox[Server])
    IN
      ∧ msg.action = "seek"
```

```
        ∧ state' = [state EXCEPT ![Server].playhead.t = msg.data.t]
        ∧ BroadcastSync

RecvSkip ≜
  ∧ inbox[Server] ≠ ⟨⟩
  ∧ LET
      server ≜ state[Server]
      msg ≜ Head(inbox[Server])
    IN
      ∧ msg.action = "skip"
      ∧ LET
          newIndex ≜ server.playhead.i + 1
          newPlayhead ≜ IF newIndex < Len(server.playlist)
                        THEN [i ↦ newIndex, t ↦ 0]
                        ELSE Stopped
        IN
          ∧ state' = [state EXCEPT ![Server].playhead = newPlayhead]
          ∧ BroadcastSync
```

Randomly lose a message from an *inbox*

```
Remove(i, seq) ≜
  [j ∈ 1 .. (Len(seq) − 1) ↦ IF j < i THEN seq[j] ELSE seq[j + 1]]

LoseMsg ≜
  ∃ n ∈ DOMAIN inbox :
    ∃ i ∈ DOMAIN inbox[n] :
      ∧ inbox' = [inbox EXCEPT ![n] = Remove(i, inbox[n])]
      ∧ UNCHANGED state
```

Spec

```
Init ≜
  ∧ inbox = [n ∈ Node ↦ ⟨⟩]
  ∧ state = [n ∈ Node ↦ InitState]

Next ≜
  ∨ ∃ self ∈ Client, song ∈ Song : SendAdd(self, song)
  ∨ ∃ self ∈ Client : RecvSync(self)
  ∨ ∃ self ∈ Client : SendSeek(self)
  ∨ ∃ self ∈ Client : SendSkip(self)
  ∨ RecvAdd
  ∨ RecvSeek
  ∨ RecvSkip
  ∨ LoseMsg
```

$Spec \triangleq$
$Init \land \Box[Next]_{vars}$

THEOREM $Spec \Rightarrow \Box TypeOK$